                                      3. Anweisungen


3._____Anweisungen

3.1.___Kommentaranweisung_REM
Format:      REM <Bemerkung> 

Zweck:       Fuegt erklaerende Bemerkungen in das Programm ein.

Bemerkungen: REM-Anweisungen  werden  nicht  ausgefuehrt.  Sie              werden nur  angezeigt,  wenn das Programm  aufge-             listet wird.

             Zu  REM-Anweisungen  kann von  einer  GOTO-  oder              GOSUB-Anweisung aus  verzweigt  werden.  Die Pro-             grammabarbeitung wird fortgesetzt mit der  ersten             durchfuehrbaren Anweisung nach der REM-Anweisung.

             Bemerkungen  koennen  auch  am Ende  einer  Zeile              angefuegt   werden,   indem  der  Bemerkung   ein              Anfuehrungszeichen  "'"  vorangeht.

             In  einer DATA-Anweisung darf REM nicht verwendet              werden.

Beispiele:   100 REM DIES IST EIN BEISPIEL                      

             110 FOR I=1 TO 15

             120 S1=S1+I

             130 NEXT I

             100 FOR I=1 TO 15  'DIES IST EIN BEISPIEL

             110 S1=S1+I

             120 NEXT I

3.2.   Typdeklarationsanweisungen

_______DEFINT_/_DEFSNG_/_DEFDBL_/_DEFSTR
Format:      DEF<Typ> <Bereich(e) der Buchstaben>,

             wobei <Typ> INT, SNG, DBL oder STR sein kann.

Zweck:       Definiert  die  Variablentypen  als   Integerzahl              (DEFINT),  Zahlen einfacher Genauigkeit (DEFSNG),              doppelter  Genauigkeit (DEFDBL) oder als Zeichen             kette (DEFSTR). 

Bemerkungen: Eine  DEF-Typ-Anweisung   legt  fest,  dass  alle                           Variablen,  beginnend mit den in <Bereich(en) der                           Buchstaben> festgelegten Buchstaben, Variable des                           spezifizierten Types sind.

             Ein  Typdeklarationszeichen   hat   immer_Vorrang              gegenueber einer DEF-Typ-Anweisung.

             Wenn  keine Typdeklarationsanweisungen  vorhanden              sind,  wird festgelegt,  dass alle Variablen ohne              Deklarationszeichen  Variable einfacher  Genauig             keit sind.

Beispiele:   10 DEFDBL L-P     Alle Variablen,  die  mit den

                               Buchstaben L,  M,  N, O und P 

                               beginnen,  sind  Variable 

                               doppelter Genauigkeit.

             10 DEFSTR A       Alle Variablen, die mit dem 

                               Buchstaben A beginnen, sind

                               Zeichenkettenvariable.

             10 DEFINT I-N,W-Z Alle Variablen, die mit den

                               Buchstaben I, J, K, L, M, N

                               und W, X, Y, Z beginnen, sind

                               Integervariable.

             10 DEFSTR A

             20 DEFINT K-Z

             30 A = "PROGRAMM"

             40 K = 10

             50 PRINT A;K

             60 END

             RUN

             PROGRAMM 10

             Ok

             In diesem Fall muss A nicht_mit_dem_$-Zeichen zur              Definition   der  Zeichenkettenvariable  versehen              werden.

3.3.___Wertzuweisungen

3.3.1._LET
Format:      [LET] <Variable>=<Ausdruck>

Zweck:       Ordnet  den Wert eines Ausdrucks einer  Variablen                           zu.

Bemerkungen: Der  Wert  kann in Form  eines  Ausdrucks,  einer              Konstanten, einer Zeichenkette oder einer anderen              Variablen desselben Typs vorgegeben sein.

             Zu beachten ist, dass das Wort LET wahlweise ver-             wendet werden kann,  d.h.  das Gleichheitszeichen              ist  ausreichend,  wenn ein Ausdruck einem  Vari             ablennamen zugeordnet wird.

             In  zeitkritischen  FOR-NEXT-Schleifen  ist  es              guenstiger, "LET" mit einzugeben.

Beispiele:      10 LET A=5      oder:      10 A=5

                20 LET B=10                20 B=10 

                30 LET C=A+B               30 C=A+B

3.3.2._SWAP
Format:      SWAP <Variable>,<Variable>

Zweck:       Tauscht die Werte von zwei Variablen aus.

Bemerkungen: Jede Typvariable  kann getauscht werden (Integer-

             zahl,  Zahl einfacher Genauigkeit, Zahl doppelter              Genauigkeit,  Zeichenkette).  Die zwei  Variablen              muessen  jedoch zum gleichen Typ gehoeren.  Sonst              erscheint ein Fehler "Type mismatch".

Beispiel:       10 A$="ABCD"

                20 B$="SIND"

                30 C$="BUCHSTABEN"

                40 PRINT A$,B$,C$

                50 SWAP A$,C$

                60 PRINT A$,B$,C$

                RUN

                ABCD        SIND      BUCHSTABEN

                BUCHSTABEN  SIND      ABCD

                0K

3.3.3._MID$
Format:      MID$(<Zeichenkettenexp1>,n[,m]) = 

                  <Zeichenkettenexp2>

             wobei  n und  m Integerausdruecke  und  

             <Zeichenkettenexp1>  sowie  <Zeichenkettenexp2> 

             Zeichenkettenausdruecke sind.

Zweck:       Ersetzt einen  Teil einer Zeichenkette durch eine                           andere Zeichenkette.

Bemerkungen: Die Zeichen in  <Zeichenkettenexp1>,  die bei der              Position n beginnen,  werden durch die Zeichen in              dem <Zeichenkettenexp2> ersetzt.

             Das wahlweise "m" bezieht sich auf die Anzahl der              Zeichen  von <Zeichenkettenexp2>,  die  beim  Er             setzen verwendet werden. 

             Wenn  "m"  weggelassen  wurde,   wird  alles  von              <Zeichenkettenexp2> verwendet.

             Ungeachtet  dessen  jedoch,  ob  "m"  weggelassen              wurde  oder enthalten ist,  geht das Ersetzen der              Zeichen  niemals  ueber  die  Originallaenge  von              <Zeichenkettenexp1> hinaus.

             MID$  ist  auch eine Funktion,  die  eine  Unter             zeichenkette  von  einer  gegebenen  Zeichenkette              zurueckbringt (siehe Funktion  MID$).

Beispiel:    10 A$ = "BERECHNEN SIE DIE FLAECHE EINES KREISES"

             20 MID$(A$,15) = "DEN UMFANG "

             30 PRINT A$

             RUN

             BERECHNEN SIE  DEN UMFANG EINES KREISES

             Ok

3.3.4._RANDOMIZE
Format:      RANDOMIZE [<Ausdruck>]

Zweck:       Stellt einen Zufallszahlengenerator neu ein.

Bemerkungen: <Ausdruck>   bestimmt  den  Anfangswert  der  Be             rechnung.

             Wenn  <Ausdruck>  fehlt,   unterbricht  BASI  die              Programmabarbeitung vor Ausfuehren der  Anweisung              RANDOMIZE. Es wird nach einem Wert durch folgende              Anzeige gefragt:

                 Random number seed (-32768 to 32767)?

             Wenn  der  Zufallszahlengenerator nicht  neu  be             schickt  wurde,  verwendet  die RND-Funktion  die              gleiche  Folge der Zufallszahlen  jedesmal  dann,              wenn das Programm abgearbeitet wird.

             Soll  die Zahlenfolge jedoch bei jeder  Programm             abarbeitung   veraendert   werden,    ist    eine              RANDOMIZE-Anweisung  am  Programmbeginn  zu  ver-             wenden.

Beispiel:    10 INPUT A

             20 RANDOMIZE

             30 FOR I= 1 TO A

             40 PRINT RND

             50 NEXT I

             60 END

             RUN

             ? 5

             Random number seed (-32768 to 32767)? 3 

                                (vom Benutzer wird 3 

                                eingegeben)

             .88598

             .484668

             .586328

             .119426

             .709225

             Ok

3.4.___Dialog-Ein-/Ausgabe

3.4.1._INPUT
Format:      INPUT[;] [<Zeichenkettenkonstante>{;|,}]

                       <Liste von Variablen>

Zweck:       Erlaubt  eine  Eingabe  waehrend der  Programmab-             arbeitung.

Bemerkungen: Wenn   eine  INPUT-Anweisung  abgearbeitet  wird,              wird  der Programmlauf unterbrochen.  Es wird ein              Fragezeichen  "?" angezeigt.  Das Programm wartet              auf die Eingabe der Daten.

             Die <Zeichenkettenkonstante> dient als  Bediener             nachricht  zur Kennzeichnung der geforderten Ein             gabe.

             Wenn eine <Zeichenkettenkonstante> vorhanden ist,              wird  diese  Zeichenkette  vor  dem  Fragezeichen              angezeigt.   Die  geforderten  Daten  werden  an             schliessend ueber die Tastatur eingegeben.

             Es  kann ein Komma anstelle eines Semikolons nach              der <Zeichenkettenkonstanten>  verwendet  werden,              um  das Fragezeichen zu unterdruecken.  Zum  Bei-             spiel bringt die Anweisung 

                 INPUT "EINFUEGEN DES GEBURTSDATUMS", B$

             die Eingabe ohne Fragezeichen.

             Wenn der Anweisung INPUT unmittelbar ein  Semiko-             lon  folgt,  dann bleibt nach <ENTER> die Kursor-             position erhalten. Die Zeilenschaltung wird nicht              ausgefuehrt.

             Die Daten,  die eingegeben werden sollen,  werden              Variablen  zugeordnet,  die in einer  <Variablen-             liste>  eingegeben  werden.  Die Anzahl  der  ge             lieferten  Datenpositionen muss die gleiche  sein              wie die Anzahl der Variablen in der Liste. Daten-             positionen werden durch Kommas getrennt.

             Die   Variablennamen   in   der   Liste   koennen              numerische oder Zeichenkettenvariablennamen (ein-             schliesslich  indizierte Variable) sein.  Der Typ              jeder Datenposition,  die eingegeben  wird,  muss              mit  dem  Typ  uebereinstimmen,   der  durch  den              Variablennamen spezifiziert wurde.

             Die  Zeichenketteneingabe bei einer  INPUT-Anwei             sung  muss nicht von Anfuehrungszeichen   umgeben              sein.

             Die Eingabe wird erst beendet,  wenn die  Variab             lenliste vollstaendig gefuellt ist.

             Werden  bei INPUT zuviel oder zuwenig Werte  oder              Werte mit einem falschen Werttyp (z.  B. Zeichen             kette  anstelle  numerisch)  eingegeben,  erfolgt              eine Fehlermeldung "?Redo from start".

             Es  wird keine Zuordnung von Eingabewerten vorge             nommen,   bis  eine  fehlerfreie Eingabe erfolgt.

Beispiele:   10 INPUT A

             20 PRINT A

             30 END

             RUN

             ? 5                   (Die 5 wurde vom Benutzer               5                    nach Anzeige des Frage-

             Ok                    zeichens eingegeben)

             10 INPUT "EINGABE BETRAG";B

             20 C = C + B

             30 PRINT "GESAMTBETRAG:";C

             40 PRINT

             50 GOTO 10

             Ok

             RUN

             EINGABE BETRAG? 9.20  (Benutzer gibt 9.20 ein.

             GESAMTBETRAG: 9.20    Zuvor wird das Frage-

             EINGABE BETRAG?       zeichen angezeigt.)

             usw.

             50 INPUT "",V         (Das Fragezeichen wird 

             60 PRINT V            unterdrueckt.)

             RUN

             10

              10

             Ok

             Weitere Beispiele:  siehe Anweisung PRINT USING

3.4.2._LINE_INPUT
Format:      LINE INPUT[;] [<Zeichenkettenkonstante>;]                    

                            <Zeichenkettenvariable>

Zweck:       Erlaubt die Eingabe einer beliebigen Zeichenkette              bis  zu 254 Zeichen (auch Komma, Leerzeichen  und              Anfuehrungszeichen moeglich).

Bemerkungen: Die  <Zeichenkettenkonstante> ist  eine  Zeichen-             kette,  die  angezeigt  wird,  bevor die  Eingabe              begonnen wurde.  Sie dient als Bedienerhinweis.

             Ein  Fragezeichen wird nur dann  angezeigt,  wenn              es Teil der Zeichenkettenkonstanten ist.

             Alle   Eingaben  vom  Ende   der  <Zeichenketten-             konstanten>  bis  <ENTER>  werden  der  <Zeichen             kettenvariablen> zugeordnet.

             Wenn  der  Anweisung LINE INPUT  unmittelbar  ein              Semikolon  folgt,  dann bleibt die Kursorposition              erhalten.  Es  wird keine Zeilenschaltung am Ende              einer Eingabezeile ausgefuehrt.

             Die  Anweisung  LINE INPUT kann durch  ^C  unter-             brochen  werden.  Das BASIC-System kehrt zum  Be             fehlsniveau zurueck und zeigt "Ok" an.

             Nach  CONT wird die Durchfuehrung bei LINE  INPUT              wieder aufgenommen.

3.4.3._PRINT
Format:
     PRINT [<Liste von Ausdruecken>]

Zweck:       Zeigt   die  im  Ergebnis  der  Abarbeitung   der              Ausdruecke  ermittelten  Werte am  Bildschirm  ab              Kursorposition an.

Bemerkungen: Wenn die <Liste von Ausdruecken> fehlt, erscheint              eine Leerzeile auf dem Bildschirm.

             Ist  eine  <Liste  von  Ausdruecken>   vorhanden,              werden  die  entsprechenden Werte der  Ausdruecke              angezeigt.

             Die  Ausdruecke in der Liste  koennen  numerische              und/oder  Zeichenkettenausdruecke sein  (Zeichen             kettenkonstante   muessen  in  Anfuehrungszeichen              eingeschlossen werden).

             Bildschirmpositionen:
             Die Position jedes angezeigten Wertes wird  durch              die  Interpunktion bestimmt,  die verwendet wird,              um die Werte in der Liste zu trennen.

             BASI unterteilt die Zeile in  Bildschirmzonen von              je 14 Stellen.

             Ein  Komma  in der <Liste von  Ausdruecken>  ver-             anlasst,  dass  der naechste Wert zu  Beginn  der              naechsten Zone angezeigt wird.

             Ein Semikolon veranlasst,  dass der naechste Wert              unmittelbar nach dem letzten Wert angezeigt wird.              Das  Eingeben von einer oder mehreren Leerstellen              zwischen den Ausdruecken hat die gleiche  Wirkung              wie das Eingeben eines Semikolons.

             In beiden Faellen wird die Anzeige der <Liste von              Ausdruecken> auf der naechsten Zeile fortgesetzt,              wenn  in  der  alten Zeile nicht  mehr  genuegend              Platz vorhanden ist.

             Beendet  ein Komma oder ein Semikolon die  <Liste              von  Ausdruecken>, beginnt  die  naechste  PRINT-             Anweisung die Anzeige auf derselben Zeile.

             Wird  die <Liste der Ausdruecke> ohne Komma  oder              Semikolon  beendet,  wird am Ende der  Zeile  ein              Zeilenvorschub  ausgefuehrt.  Der Kursor befindet             sich am Anfang der neuen Zeile.

             Gedruckten  / angezeigten Zahlen folgt immer  ein              Leerraum (Space). Positiven Zahlen geht ein Leer-             raum voran, negativen Zahlen ein Minuszeichen.

             Es  kann  ein Fragezeichen  anstelle  des  Wortes              PRINT  in einer Print-Anweisung verwendet werden.

Beispiel 1:  10 W=50

             20 PRINT W, W+10, W*10

             30 PRINT W; W+10; W*10

             40 END

             RUN

              50           60           500           

              50  60  500

             Ok

             In  diesem Beispiel veranlassen die Kommas in der              PRINT-Anweisung,  dass  jeder Wert am Beginn  der              naechsten Druckzone angezeigt wird, und das Semi-             kolon,  dass  die Werte hintereinander  angezeigt              werden.

Beispiel 2:  10 FOR I = 1 TO 5

             20 A = A + 5

             30 M = M + 10

             40 ? A; M;

             50  NEXT I

             Ok

             RUN

              5  10   10   20   15   30   20   40  25
50

             Ok

             In  Zeile  40 wird ein Fragezeichen anstelle  des              Wortes PRINT verwendet.

3.4.4._PRINT_USING
Format:      PRINT USING <Zeichenkettenexp>;

                                 <Liste von Ausdruecken>

Zweck:       Zeigt Zeichenketten  oder Zahlen unter Verwendung                 eines spezifizierten Formates am Bildschirm an.

Bemerkungen  <Liste von Ausdruecken> ist aus Zeichenkettenaus-

und          druecken  oder numerischen Ausdruecken  zusammenBeispiele:   gesetzt,  die   angezeigt  werden  sollen,  durch              Semikolons getrennt.

             <Zeichenkettenexp> ist ein Zeichenkettenausdruck,              der aus speziellen Formatzeichen  zusammengesetzt              ist.  Diese  Formatzeichen bestimmen das Feld und              das  Format der anzuzeigenden Zeichenketten  oder              Zahlen.

                    Zeichenkettenfelder:
             Wenn PRINT USING verwendet wird, um Zeichenketten              anzuzeigen,  kann  eines von  drei  Formatzeichen              verwendet  werden,  um  das Zeichenkettenfeld  zu              formatieren:

  "!"        Spezifiziert,  dass  nur  das erste  Zeichen  der              angegebenen Zeichenkette angezeigt werden soll.

             Beispiel:  10 LET A$="ABCDEF"

                        20 LET B$="1250"

                        30 PRINT USING "!";A$,B$

                        40 END

                        RUN

                        A1

"\n Leerzeichen\"

             Spezifiziert, dass 2 + n Zeichen der Zeichenkette              angezeigt werden sollen. 

             Wenn die  Schraegstriche ohne Leerraum eingegeben              werden,  werden zwei Zeichen angezeigt, mit einem              Leerraum werden drei Zeichen angezeigt usw.

             Wenn  die Zeichenkette laenger als das Feld  ist,              werden die restlichen Zeichen ignoriert. 

             Wenn  das Feld laenger als die Zeichenkette  ist,              wird die Zeichenkette linksbuendig angeordnet und              bis zur spezifizierten Laenge rechts durch  Leer             zeichen aufgefuellt.

             Beispiel:  10 LET A$="ABCDEF"

                        20 LET B$="1250"

                        30 PRINT USING "\  \";A$,B$

                        40 END

                        RUN

                        ABCD1250

"&"          Spezifiziert  ein Zeichenkettenfeld mit variabler              Laenge.  Wenn das Feld mit "&" spezifiziert wird,              wird  die  Zeichenkette ohne Veraenderung  ausge             geben.

             Beispiel:  10 LET A$="ABCDEF"

                        20 LET B$="1250"

                        30 PRINT USING "&";A$,B$

                        40 END

                        RUN

                        ABCDEF1250

                          Numerische_Felder
             Wenn PRINT USING verwendet wird,  um Zahlen anzu             zeigen,  koennen die folgenden speziellen Zeichen              verwendet werden,  um das numerische Feld format             maessig aufzuteilen:

"#"          Es  wird  ein Nummernzeichen verwendet,  um  jede              Ziffernposition  darzustellen.  Die  Ziffernposi             tionen werden immer gefuellt.  

             Wenn  die anzuzeigende Zahl weniger Ziffern  hat,              als Positionen spezifiziert wurden, wird die Zahl              rechtsbuendig  angeordnet (wobei Leerzeichen vor             angehen).

             Werden  weniger  Positionen   spezifiziert,   als              Ziffernpositionen  vorhanden sind,  wird die  ge-             samte Zahl angezeigt,  und vor der Zahl steht ein              Zeichen "%".

             Beispiel:  10 LET A=125.25

                        20 LET B=.75

                        30 PRINT USING "##.##  ";A,B

                        40 END

                        RUN

                        %125.25     0.75

"."          Ein  Dezimalpunkt kann an jeder beliebigen  Posi             tion  im  Feld  eingesetzt  werden.   

             Wenn  die Formatzeichenkette  spezifiziert,  dass              einer  Ziffer der Dezimalpunkt  vorangehen  soll,              wird die Ziffer immer angezeigt (als 0, wenn not-             wendig). 

             Zahlen werden, wenn erforderlich, gerundet.

             Beispiel:  10 LET A=125.25

                        20 LET B=.75

                        30 PRINT USING "##.#";B

                        40 END

                        RUN

                         0.8

                        10 A=125.25

                        20 B=.75

                        30 PRINT USING "##.##  ";B

                        40 END

                        RUN

                         0.75

"+"          Ein  Pluszeichen  am  Beginn  oder  am  Ende  der              Formatzeichenkette  veranlasst,   dass  das  Vor             zeichen  der Zahl (plus oder minus) vor oder nach              der Zahl angezeigt wird.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "+##.##  ";A,B,C

                        120 END

                        RUN

                        -12.50   +0.00  -80.50

"-"          Ein  Minuszeichen am Ende  der Formatzeichenkette              veranlasst,  dass negative Zahlen mit einem nach-             folgenden Minuszeichen angezeigt werden.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "##.##-  ";A,B,C

                        120 END

                        RUN

                        12.50-  0.00   12.50

"**"         Ein doppelter Stern zu Beginn der  Formatzeichen             kette  veranlasst,  dass vorangehende Leerstellen              im numerischen Feld mit Stern gefuellt werden.

             Die  "**" spezifizieren auch die Positionen  fuer              zwei weitere Ziffern.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "**##.##  ";A,B,C

                        120 END

                        RUN

                        *129.46  1234.00  ***0.96

                        ***9.45  2345.00  **95.00

"$$"         Ein Doppel-Waehrungszeichen veranlasst,  dass ein              Waehrungszeichen unmittelbar links von der forma             tierten Zahl angezeigt wird. 

             Das  "$$" spezifiziert zwei weitere  Ziffernposi             tionen,   von  denen  eine  das  Waehrungszeichen              erhaelt. 

             Das  Exponentialformat kann bei "$$"  nicht  ver             wendet werden.

             Negative   Zahlen  koennen  nur  dann   verwendet              werden, wenn das Minuszeichen rechts nachfolgt.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "$$##.##  ";A,B,C

                        120 END

                        RUN

                          $9.56   $12.00  $456.50

"**$"        Das   "**$"  zu  Beginn  der   Formatzeichenkette              kombiniert die Wirkungen der obigen zwei Symbole. 

             Vorangehende   Leerstellen   werden   mit   Stern              gefuellt.  Ein Waehrungszeichen wird vor der Zahl              angezeigt.

             "**$"  speizifiziert  drei  weitere  Ziffernposi-             tionen,  von denen eine das Waehrungszeichen ent             haelt.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "**$##.##  ";A,B,C

                        120 END

                        RUN

                        ***$9.56  **$25.00  %$12345.00

                        **$95.00  *-$25.25  $1234.00

","          Ein  Komma,  das  sich links vom  Dezimalpunkt in              einer  formataufteilenden Zeichenkette  befindet,              veranlasst, dass links vom Dezimalpunkt ein Komma              links  von  jeder dritten Ziffer  angezeigt  wird

             (Tausenderinterpunktion).

             Ein Komma,  das sich am  Ende der  Formatzeichen             kette  befindet,  wird als Teil der  Zeichenkette              angezeigt. 

             Ein     Komma    spezifiziert    eine     weitere              Ziffernposition.  

             Das  Komma  hat keine Wirkung,  wenn es  mit  dem              Exponentialformat (^^^^) verwendet wird.

             Beispiel:  100 INPUT A,B,C

                        110 PRINT USING "####,.##  ";A,B,C

                        120 END

                        RUN

                        1,234.00    256.00  4,561.25

                        100 INPUT A,B,C

                        105 PRINT A,B,C

                        110 PRINT USING "##.##,  ";A,B,C

                        120 END

                        RUN

                         12.25         12.456        789

                        12.25,  12.46,  89.00,

"^^^^"       Vier ^-Zeichen oder Aufwaertspfeile koennen  nach              den Ziffernpositionszeichen plaziert  werden,  um              das Exponentialformat zu spezifizieren.

             Die vier Aufwaertspfeile halten den Leerraum fuer              E+xx frei.

             Es  kann  jede Dezimalpunktposition  spezifiziert              werden.

             Die  signifikanten Ziffern werden links  justiert              und der Exponent eingestellt.

             Wenn nicht ein vorangehendes "+" oder nachfolgen             des  "+" oder "-" spezifiziert  worden ist,  wird              eine  Ziffernposition links vom Dezimalpunkt ver             wendet,  um  einen Leerraum oder ein Minuszeichen              auszugeben.

             Beispiel:  100 INPUT A,B,C

                        105 PRINT A,B,C

                        110 PRINT USING "##.##^^^^  ";A,B,C

                        120 END

                        RUN

                         1234.25       1.45678       .789456

                         1.23E+03   1.46E+00   7.89E-01

                        100 INPUT A,B,C

                        105 PRINT A,B,C

                        110 PRINT USING ".##^^^^  ";A,B,C

                        120 END

                        RUN

                         12345.6      -45.25         12789

                        .12E+05   .45E+02-  .13E+05

%            Wenn die anzuzeigende Zahl groesser als das  spe             zifizierte numerische Feld ist, wird ein Prozent             zeichen vor der Zahl ausgegeben. 

             Wenn das Runden der Grund ist,  dass die Zahl das              Feld ueberschreitet,  wird ein Prozentzeichen vor              der gerundeten Zahl ausgegeben.

             Beispiel:  10 LET A=125.25

                        20 LET B=.75

                        30 PRINT USING "#.###  ";A,B

                        40 END

                        RUN

                        %125.250  0.750

             Wenn  die  Anzahl der spezifizierten  Ziffern  24              ueberschreitet, wird ein Fehler 

                  "Illegal function call"   

             angezeigt.

Anmerkung:   Leerzeichen  in der Formatzeichenkette  bedeuten,              dass diese bei der Ausgabe mit erscheinen.

3.4.5._LPRINT und LPRINT_USING
Format:      LPRINT [<Liste von Ausdruecken>]

             LPRINT USING <Zeichenkettenexp>;<Liste von 

                                             Ausdruecken>

Zweck:       Ausgabe von Daten ueber Drucker.

Bemerkungen: Es  sind die gleichen Bemerkungen wie  bei  PRINT              und PRINT USING gueltig.  Die Ausgabe erfolgt zum              Drucker.

             LPRINT  nimmt einen Drucker mit einer Breite  von              132 Zeichen an.

3.4.6._WRITE
Format:      WRITE[ <Liste von Ausdruecken>]

Zweck:       Ausgabe von Daten ueber Bildschirm.

Bemerkungen: Wenn die <Liste von Ausdruecken> fehlt, wird eine              Leerzeile ausgegeben.

             Wenn  die <Liste von Ausdruecken> enthalten  ist,              werden die Werte der Ausdruecke ausgegeben.

             Die  Ausdruecke  in der Liste koennen  numerische              und/oder   Zeichenkettenausdruecke   sein.    Sie              muessen durch Kommas getrennt werden.

             Bei der Ausgabe wird jede Position von der  letz             ten durch ein Komma getrennt.

             Zeichenketten  werden  durch   Anfuehrungszeichen              begrenzt. 

             Nachdem   die   letzte  Position  in  der   Liste              angezeigt  wurde,  fuehrt das BASIC-System  einen              Zeilenvorschub aus,  und der Kursor befindet sich              am Anfang der Zeile.

Beispiel:    10 A=50 : B=70 : C$="WEITERE EINGABEN"

             20 WRITE A,B,C$

             RUN

              50, 70,"WEITERE EINGABEN"

             Ok

3.4.7._WIDTH
Format:      WIDTH [LPRINT] <Integerausdruck>

Zweck:       Einstellen  der  Laenge der Druckzeile  fuer  den              Bildschirm  oder Drucker.

Bemerkungen: Fehlt die Angabe LPRINT, wird die Zeilenbreite am              Bildschirm eingestellt.

             Bei   Angabe  LPRINT  wird  die  Zeilenbreite  am              Drucker eingestellt.

             Der  <Integerausdruck> muss ein Wert  im  Bereich              von 15 ... 255 sein.

             Fehlt diese Angabe, werden 72 Zeichen angenommen. 

             Wenn der <Integerausdruck> 255 betraegt,  ist die              Zeilenbreite "unendlich",  das heisst, das BASIC-             System fuehrt  keinen Wagenruecklauf aus.

             Jedoch  wird  bei  Nutzung  der  POS- oder  LPOS-             Funktion  die  Position  des  Kursors oder Druck-

             kopfes nach der Position 255 zu Null.

Beispiel:         10 PRINT "ABCDEFGHIJKLMNOPQRSTUVWXYZ"

                  RUN

                  ABCDEFGHIJKLMNOPQRSTUVWXYZ

                  Ok

                  WIDTH 10

                  Ok

                  RUN

                  ABCDEFGHIJ        

                  KLMNOPQRST

                  UVWXYZ

                  Ok

3.5._Arbeit_mit_dem_Konstantenspeicher

3.5.1._READ
Format:      READ <Liste von Variablen>

Zweck:       Lesen  von  Werten aus einer  DATA-Anweisung  und              Zuweisung an Variable.

Bemerkungen: Eine READ-Anweisung muss immer in Verbindung  mit              einer  DATA-Anweisung  verwendet  werden.   

             READ-Anweisungen ordnen den DATA-Anweisungswerten              Variable  zu.  Die Variablen  der  READ-Anweisung              koennen   numerische  oder  Zeichenkettenvariable              sein.

             Die gelesenen Werte muessen mit den  spezifizier             ten  Variablentypen  uebereinstimmen.   Wenn  sie              nicht   uebereinstimmen,   erscheint  ein  Fehler              "Syntax error".

             Eine  einzige READ-Anweisung kann auf  eine  oder              mehrere  DATA-Anweisungen  zugreifen (sie  werden              der  Reihe nach abgearbeitet) oder mehrere  READ-             Anweisungen  koennen auf dieselbe  DATA-Anweisung              zugreifen.

             Wenn  die Anzahl der Variablen in der <Liste  von              Variablen>  die Anzahl der Elemente in der  (den)              DATA-Anweisung(en) uebersteigt,  wird ein  Fehler              "Out of data" angezeigt.

             Ist  die Anzahl der spezifizierten Variablen  ge             ringer  als die Anzahl der Elemente in der  (den)              DATA-Anweisung(en),  beginnen  die  nachfolgenden              READ-Anweisungen  mit  dem Lesen von  Daten  beim              ersten ungelesenen Element. 

             Gibt  es  keine  nachfolgenden  READ-Anweisungen,             werden die restlichen Daten ignoriert.

             Um DATA-Anweisungen vom Beginn neu zu lesen, wird              die RESTORE-Anweisung verwendet.

Beispiel:    10 READ A,B,D$                 

             20 DATA 10,20,"ABCD"

             30 PRINT A,B,D$,A

             40 PRINT A+B,A-B

             50 PRINT "BEISPIEL: "D$

             60 END

             RUN

              10            20            ABCD          10

              30           -10

             BEISPIEL: ABCD

             Ok

3.5.2._DATA
Format:      DATA <Liste der Konstanten>

Zweck:       Angabe  der  numerischen   und  Zeichenkettenkon-             stanten,  die  durch  die READ-Anweisung(en)  des              Programmes gelesen werden sollen (siehe Anweisung              READ).

Bemerkung:   DATA-Anweisungen   sind  nicht  ausfuehrbar   und              koennen  irgendwo im Programm angeordnet  werden. 

             Eine  DATA-Anweisung kann soviel Konstanten  ent             halten,  wie auf eine Zeile passen (durch  Kommas              getrennt).

             Es  koennen eine beliebige Anzahl von DATA-Anwei-             sungen  in einem Programm verwendet  werden.  Die              READ-Anweisungen  uebernehmen die Werte der DATA-             Anweisungen  der  Reihe  nach,  entsprechend  der              Anordnung im Programm.

             <Liste der Konstanten> kann numerische Konstanten              in jedem beliebigen Format enthalten,  das heisst              Festkomma,  Gleitkomma oder Integerzahl.  In  der              Liste sind keine  numerischen Ausdruecke erlaubt.

             Zeichenkettenkonstante     in    DATA-Anweisungen              muessen nur dann von doppelten Anfuehrungszeichen              umgeben sein,  wenn sie Kommas, Doppelpunkte oder              bedeutsame vorangestellte oder nachfolgende Leer-             stellen enthalten.  Sonst sind Anfuehrungszeichen              nicht erforderlich.

             Der  Variablentyp  (numerisch oder Zeichenkette),              der in der READ-Anweisung angeben wird,  muss mit              der entsprechenden Konstanten in der  DATA-Anwei-             sung uebereinstimmen.

             DATA-Anweisungen  koennen neu gelesen werden  vom              Beginn an unter Verwendung der RESTORE-Anweisung.

Beispiel:    siehe Anweisung READ

3.5.3._RESTORE
Format:      RESTORE [<Zeilennummer>]

Zweck:       Dient  der  Positionierung auf das erste  Element              der DATA-Anweisung, deren Zeilennummer  angegeben              wurde.

Bemerkungen: Ist  keine Zeilennummer angegeben,  wird auf  das              erste   Element   der  DATA-Anweisung   mit   der              niedrigsten Anweisungsnummer positioniert.

Beispiel:    10 READ A,B,C

             20 RESTORE

             30 READ D,E,F

             40 DATA 50, 60, 70

             50 PRINT A;B;C;D;E;F

             RUN

              50  60  70  50  60  70

             Ok

3.6.___Steueranweisungen

3.6.1._STOP
Format:      STOP

Zweck:       Beenden der Programmabarbeitung und Rueckkehr zum                           Befehlsniveau.

Bemerkungen: STOP-Anweisungen   koennen   irgendwo  in   einem                           Programm verwendet werden,  um die Ausfuehrung zu              beenden.

             Wenn  ein STOP erreicht wird,  wird die  folgende              Nachricht angezeigt:

                               Break in nnnnn

             Im Gegensatz zur END-Anweisung werden die Dateien              mit der STOP-Anweisung nicht geschlossen.

             Nach  einer STOP-Anweisung kehrt das BASIC-System              immer zum Befehlsniveau zurueck.

             Die  Programmabarbeitung  kann  mit  einer  CONT-             Anweisung wieder aufgenommen werden.

Beispiel:       10 INPUT A 

                20 B=2.5

                30 C=6.9

                40 SUM=B*A+(A/C)

                50 STOP

                60 PRINT SUM

                70 SUM=SUM+1

                80 GOTO 10

                RUN

                ? 10

                Break in 50

                Ok

                PRINT A

                 10

                Ok

                CONT

                 26.4493

                ?

3.6.2._END
Format:      END

Zweck:       Beenden der Programmabarbeitung, Schliessen aller              Dateien  und  Rueckkehr  zum  Befehlsniveau   des              BASIC-Systems.

Bemerkungen: END-Anweisungen  koennen  irgendwo  im   Programm               angeordnet werden,  um das Abarbeiten zu beenden.              Anders  als  die  STOP-Anweisung  veranlasst  END              nicht    die   Anzeige   einer   BREAK-Nachricht.

             Eine  END-Anweisung  am Ende eines Programms  ist              wahlweise  moeglich.  

             Das  BASIC-System  kehrt immer zum  Befehlsniveau              zurueck, nachdem ein END abgearbeitet worden ist.

Beispiel:    200 IF I>500 THEN END ELSE GOTO 50 

3.6.3._GOTO
Format:      GOTO <Zeilennummer>

Zweck:       Unbedingte Verzweigung aus der normalen Programm             folge zu einer spezifizierten Zeilennummer.

Bemerkungen: Wenn <Zeilennummer> eine durchfuehrbare Anweisung              ist,  wird das Programm ab dieser Anweisung fort             gesetzt.

             Wenn  es eine nichtdurchfuehrbare Anweisung  ist,              geht  die Ausfuehrung an der naechsten  folgenden              durchfuehrbaren Anweisung weiter.

3.6.4._GOSUB_..._RETURN
Format:      GOSUB <Zeilennummer>

Zweck:       Verzweigen  zu  einem  Unterprogramm  und  Rueck             kehr.

Bmerkungen:  <Zeilennummer>  ist  die erste Zeile  des  Unter-             programmes.                          Ein  Unterprogramm  kann beliebig  oft  in  einem              Programm gerufen werden. 

             Die   RETURN-Anweisung  in  einem   Unterprogramm              veranlasst  das  BASIC-System,  zu der  Anweisung              zurueckzukehren,  die der GOSUB-Anweisung  folgt,              von  der  aus zum Unterprogramm verzweigt  wurde.              Ein  Unterprogramm  kann mehr  als  eine  RETURN-             Anweisung enthalten. 

             Ein  Unterprogramm  kann innerhalb eines  anderen              Unterprogramms  gerufen werden.  Solch eine  Ver             schachtelung  von Unterprogrammen wird nur  durch              den zur Verfuegung stehenden Speicher begrenzt.

             Unterprogramme koennen irgendwo im Programm  ste             hen.  Es  wird  empfohlen,  Unterprogramme so  zu              schreiben,  dass sie vom Hauptprogramm leicht  zu              unterscheiden sind.

Beispiel:    10 GOSUB 40

             20 PRINT "RUECKSPRUNG AUS UP"

             30 END

             40 PRINT "Abarbeiten  ";

             50 PRINT "Unterprogramm"

             60 RETURN

             RUN

             Abarbeiten Unterprogramm

             RUECKSPRUNG AUS UP

             0k

3.6.5._ON_..._GOSUB und ON_..._GOTO
Format:      ON <Ausdruck> GOTO <Liste der Zeilennummern>

             ON <Ausdruck> GOSUB <Liste der Zeilennummern>

Zweck:       Verzweigen  zu  einer der verschiedenen  spezifi             zierten  Zeilennummern in  Abhaengigkeit vom Wert              des Ausdruckes.

Bemerkungen: Der   Wert  des  <Ausdruckes>  bestimmt,   welche                        Zeilennummer  in der Liste als Sprungziel verwen             det wird.

             Die Sprungziele werden ausgewaehlt, indem sie bei              1 beginnend durchnumeriert werden. 

             Es wird folgendermassen ausgewaehlt:

                 Die 1. Zeilennummer, wenn der Wert 

                 des Ausdruckes  = 1  ist.

                 Die 2. Zeilennummer, wenn der Wert 

                 des Ausdruckes = 2 ist usw.

             Wenn der Wert des <Ausdruckes> Null oder groesser              ist  als die Anzahl der Sprungziele in der  Liste              (aber  kleiner  oder  gleich  255),   wird  nicht              verzweigt.  Das BASIC-System setzt mit der naech             sten durchfuehrbaren Anweisung fort. 

             Wenn der  Wert  des  <Ausdruckes>  negativ   oder              groesser  als  255  ist,   erscheint  ein  Fehler 

                    "Illegal function call".

             Wenn  der Wert keine Integerzahl  ist,  wird  der              Bruchteil gerundet.

             In  der  ON...GOSUB-Anweisung muss  jede  Zeilen             nummer in der Liste die erste Zeilennummer  eines              Unterprogramms sein.

             Die Sprungziele sind durch Komma zu trennen.

Beispiel:    10 LET I=0

             20 LET I=I+1

             30 ON I GOTO 50,70,90

             35 PRINT "ABBRUCH"

             40 END

             50 PRINT "I= "I

             60 GOTO 20

             70 PRINT "I= "I

             80 GOTO 20

             90 PRINT "I= "I

             100 GOTO 20

             RUN

             I=  1

             I=  2

             I=  3

             ABBRUCH

             Ok

3.6.6._FOR_..._NEXT
Format:      FOR <Variable> = x TO y [STEP z]             

             .

             .

             .

             NEXT [<Variable>] [,<Variable>...]

             wobei x, y und z numerische Ausdruecke sind.

Zweck:       Mehrfaches  Abarbeiten  einer  Folge  von  Anwei             sungen.

Bemerkungen: <Variable> wird als Zaehler verwendet.  

             Der  erste  numerische Ausdruck (x) ist  der  An             fangswert  des Zaehlers.  Der  zweite  numerische              Ausdruck (y) ist der Endwert des Zaehlers. 

             Die Programmzeilen, die der FOR-Anweisung folgen,              werden  ausgefuehrt,  bis  das Programm  auf  die              NEXT-Anweisung trifft.  Dann nimmt der Zaehler zu              um den Wert,  der durch STEP spezifiziert  worden              ist.  Es wird eine Kontrolle durchgefuehrt, um zu              sehen,  ob  der Wert des Zaehlers jetzt  groesser              als der Endwert (y) ist.              

             Ist  der  Wert nicht groesser,  geht  das  BASIC-             System  zurueck  zu  der Anweisung nach der  FOR-             Anweisung. Der Prozess wird wiederholt.  

             Wenn  er  groesser ist,  wird mit  der  Anweisung              fortgesetzt,  die der NEXT-Anweisung folgt.  Dies              ist ein FOR...NEXT-Zyklus.

             Wurde  STEP  nicht  spezifiziert,  wird  mit  der              Schrittweite =1 gearbeitet.

             Ist STEP negativ,  wird der Endwert des  Zaehlers              so eingestellt,  dass er kleiner als der Anfangs             wert  ist.  Der Zaehler wird jedesmal verringert,              wenn  der Zyklus abgearbeitet  wird.  Der  Zyklus              wird  durchgefuehrt,  bis der Zaehler kleiner als              der Endwert ist.

             Der Zaehler muss eine numerische Integerkonstante              oder Konstante einfacher Genauigkeit  sein.  

             Wird eine numerische Konstante doppelter Genauig             keit  verwendet,  ergibt das einen  Fehler  "Type              mismatch".

             Verschachtelte_Zyklen

     FOR...NEXT-Zyklen  koennen verschachtelt  werden,              d.h.  ein FOR...NEXT-Zyklus kann innerhalb  eines              anderen FOR...NEXT-Zyklus enthalten  sein.   

             Werden Zyklen  verschachtelt,  muss jeder  Zyklus              einen   alleinigen  Variablennamen  als   Zaehler              haben.  Die  NEXT-Anweisung fuer den  Innenzyklus              muss  vor jener fuer den Aussenzyklus  auftreten. 

             Haben verschachtelte Zyklen  denselben  Endpunkt,             kann  fuer alle eine einzige  NEXT-Anweisung ver             wendet werden.

             Die  Variable(n)  in der  NEXT-Anweisung  koennen              weggelassen  werden.  In diesem Fall werden diese                          NEXT-Anweisung   und  die  letzte   FOR-Anweisung                           als zusammengehoerig betrachtet.

             Wenn  eine NEXT-Anweisung vor der  entsprechenden              FOR-Anweisung auftritt, wird eine Fehlernachricht                "NEXT without FOR" ausgegeben, und die Durchfueh             rung wird beendet. 

             Es  ist zu empfehlen,  keine  REM-Anweisungen  in              FOR...NEXT-Schleifen  zu verwenden,  um die  Pro             grammlaufzeiten nicht zu erhoehen.

Beispiel 1:  10 INPUT X

             20 FOR I=X TO 20 STEP 2

             30 PRINT I

             40 NEXT I

             RUN

             ? 2    (Eingabe: 2)

              2

              4

              6

              8

              10

              12

              14

              16

              18

              20

             0k

Beispiel 2:  10 J=0

             20 FOR I=1 TO J

             30 PRINT I

             40 NEXT I

             In diesem Beispiel wird der Zyklus nicht durchge             fuehrt,  weil  der Anfangswert des  Zyklusses den              Endwert  uebersteigt.

Beispiel 3:  FOR...NEXT kann auch fuer Warteschleifen Ver-             wendung finden:

             z.B:   30 FOR D% = 0 TO 1000 : NEXT

                          - # -


