                                    5. Dateiarbeit


5._Dateiarbeit:_Anweisungen_und_Funktionen

5.1._Allgemeines
Es kann mit 2 verschiedenen Dateien gearbeitet werden:

                    - Sequentielle Dateien

                    - Direktzugriffsdateien

Sequentielle__Dateien  sind leichter zu schaffen  als  Direkt-zugriffsdateien.

Wenn aber wieder auf die Dateien zugegriffen werden soll, sind sequentielle  Dateien  in  ihrer Flexibilitaet  und  Verarbeitungsgeschwindigkeit eingeschraenkt.

Die  Daten,  die  in eine  sequentielle  Datei  eingeschrieben werden  sollen,  werden  Satz fuer Satz (sequentiell) auf  der Diskette gespeichert.

Sie werden ebenfalls sequentiell von der Diskette wieder  ein-gelesen.

Eine sequentielle Datei  wird  als  ASCII-Zeichen gespeichert.

Das  Schaffen und der Zugriff auf  Direktzugriffs-_(wahlfreie) Dateien  erfordern mehr Programmschritte als bei  seqentiellen Dateien,  aber es entstehen bei Verwendung von Direktzugriffs-dateien Vorteile.

Ein Vorteil ist,  dass Direktzugriffsdateien weniger Platz auf der Diskette benoetigen,  weil das BASIC-System diese in einem verdichteten Binaerformat speichert.

Der  groesste Vorteil der Direktzugriffsdateien ist,  dass direkt  ueber die Satznummer auf die entsprechenden Daten  zugegriffen werden kann.

5.2._Eroeffnen_und_Schliessen_von_Dateien
5.2.1._OPEN
Format:      OPEN <"Modus">, [#]<Dateinummer>, <"Dateiname">

             [,<Satzlaenge>]

Zweck:       Eroeffnet eine Diskettendatei. Erst nach dem OPEN              ist ein lesender / schreibender Zugriff auf diese              Datei moeglich.

Bemerkungen: OPEN  ordnet einen Puffer fuer Eingabe /  Ausgabe              der Datei zu und bestimmt den Zugriffsmodus,  der              zum Puffer verwendet wird.

             <Modus>  kann  eine der  folgenden  Zugriffsarten              sein:

               O  spezifiziert die sequentielle Ausgabe                   (Schreiben)

               I  spezifiziert die sequentielle Eingabe

                  (Lesen)

               R  spezifiziert die Ein-/ Ausgabe im Direkt-                  zugriff     

             <Dateinummer>   ist  ein  ganzzahliger   Ausdruck              zwischen 1 und 15.

             Die  Zahl ist solange fuer die Datei  reserviert,              wie  die Datei offengehalten wird.  Sie wird ver-             wendet,  um andere Disketten- Ein-/  Ausgabe- An             weisungen auf diese Datei zu beziehen.

             <Dateiname>  ist ein  Zeichenkettenausdruck,  der              den  Namen  der Datei enthaelt.

             <Satzlaenge> ist ein ganzzahliger Ausdruck,  der,              falls  er enthalten ist,  die Aufzeichnungslaenge              fuer Direktzugriffsdateien festlegt.  Fehlt diese              Angabe,  ist  die  Direktzugriffsdatei  128  Byte              lang.

             Bei  sequentiellen Dateien wird keine  Satzlaenge              angegeben.  Es wird mit fester Satzlaenge von 128              Byte gearbeitet.

Anmerkung:   Bei  der  sequentiellen Eingabe oder  im  Direkt             zugriff kann eine Datei mit mehreren Dateinummern              auf einmal eroeffnet werden.

             Bei der sequentiellen Ausgabe kann eine Datei mit              nur einer Dateinummer eroeffnet werden.

             Die  Angabe von <Modus> sowie <Dateiname> hat  in              Anfuehrungszeichen zu erfolgen.

Beispiel:    10 OPEN "I",#1,"PROG"

5.2.2._CLOSE
Format:      CLOSE [[#]<Dateinummer> [,[#]<Dateinummer...>]]

Zweck:       Beendet  die  Ein- / Ausgabe zu einer  Disketten             datei. 

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              eroeffnet wurde (OPEN).

             Ein  CLOSE ohne Argumente schliesst alle  offenen              Dateien.

             Die Verbindung zwischen einer einzelnen Datei und              der  Dateinummer endet bei der Ausfuehrung  eines              CLOSE.

             Die  Datei  kann  wiedereroeffnet  werden,  indem              dieselbe  oder eine andere Dateinummer  verwendet              wird. Ebenfalls kann diese Dateinummer wiederver-             wendet werden,  um  irgeneine andere Datei zu er-             oeffnen.

             Eine  Datei  kann erst wieder  eroeffnet  werden,              wenn sie vorher geschlossen wurde.

             Die  END-Anweisung und das NEW-Komando schliessen              immer  alle Dateien automatisch  (STOP  schliesst              keine Dateien).

5.3._Sequentielle_Dateien
Die Anweisungen und Funktionen,  die bei sequentiellen Dateien verwendet werden, sind:

     PRINT #

     PRINT # USING

     WRITE #

     INPUT #

     LINE INPUT #

     EOF

     LOC

Die  folgenden  Programmschritte sind  erforderlich,  um  eine sequentielle_Datei__zu_schaffen  und  darauf  zuzugreifen:

                                           Beispiel:
1. OPEN (Eroeffnen) der Datei im           OPEN "O",#1,"DATEN"

   "O"-Modus (sequentielle Ausgabe)

2. Schreiben von Daten in die Datei        PRINT #1,A$;B$;C$ 

   mit  PRINT #.

   Es koennen auch  WRITE #  oder 

   PRINT # USING  benutzt werden.

3. Um auf die Daten in der Datei           CLOSE #1

   zuzugreifen, muss man die Datei  

   mit CLOSE schliessen und mit            OPEN "I",#1,"DATEN"

   OPEN im "I"-Modus wieder er-

   oeffnen (sequentielle Eingabe).

4. Die INPUT #-Anweisung ist zum           INPUT #1,X$,Y$,Z$

   Einlesen der Daten aus der 

   sequentiellen Datei zu ver-

   wenden. Es kann auch 

   LINE INPUT #  benutzt werden.

5. Um einen Fehler "Input past end"        IF EOF(1) THEN END

   zu vermeiden, ist die EOF-Funktion  

   zu nutzen, welche das Ende der 

   Datei testet.

6. Die LOC-Funktion ermittelt die          IF LOC(1) > 100

   Anzahl der Sektoren, die in die         THEN GOTO 200

   Datei eingeschrieben oder aus der 

   Datei gelesen wurden (128-Byte-

   Datenblock).

Soll  eine bereits existierende sequentielle_Datei spaeter  an ihrem Ende erweitert werden,  so ist es nicht moeglich,  diese Datei  einfach im O-Modus zu eroeffnen und anschliessend fortzuschreiben.

Sobald  man  eine  sequentielle Datei  im  O-Modus  eroeffnet, zerstoert man ihren aktuellen Inhalt.

Es ist deshalb wie folgt zu verfahren:

Es  wird  mit 2 Dateien gearbeitet.  Die bereits  existierende Datei  wurde  unter  dem  <Dateinamen> "DATEN"  aufgezeichnet.

Fuer die  2. Datei  wurde  der  Name "COPY" gewaehlt.

1.  Eroeffnen der Datei "DATEN" im "I"-Modus (Input).

2.  Eroeffnen der Datei "COPY"  im "O"-Modus (Output).

3.  Einlesen der Daten aus der Datei "DATEN".

4.  Einschreiben der Daten in die Datei "COPY".

5.  Schliessen der Datei "DATEN" und Loeschen (KILL) "DATEN".

6.  Einschreiben der neuen Informationen in die Datei "COPY".

7.  Umbenennen (NAME) von "COPY" in "DATEN" und schliessen der     Datei "DATEN".

Jetzt  befindet  sich eine  erweiterte Datei "DATEN"  auf  der Diskette,  die  alle  Daten  enthaelt.

Alle  Details sind den  jeweiligen Anweisungen oder Funktionen bzw. Beispielen zu entnehmen.

5.3.1._PRINT_#__und__PRINT_#_USING
Format:      PRINT #<Dateinummer>, [USING <Zeichenkettenexp>;]

             <Liste der Ausdruecke>

Zweck:       Ausgabe von Daten in eine sequentielle Disketten             datei. 

Bemerkungen: <Dateinummer> ist die Zahl,  die verwendet  wird,              wenn die Datei fuer die Ausgabe eroeffnet wurde.

             <Zeichenkettenexp>  ist aus den Formatzeichen zu-             sammengesetzt,  die in der Anweisung PRINT  USING              beschrieben wurden.

             Die  Ausdruecke  in <Liste der  Ausdruecke>  sind              numerische  und  / oder  Zeichenkettenausdruecke,              welche in die Datei geschrieben werden.

             PRINT # und PRINT # USING schreiben die Daten  in              einen  Ausgabepuffer.  Ausgegeben   auf  Diskette              werden  die Daten erst,  wenn der Puffer gefuellt              ist  oder bei einem CLOSE (bzw. END oder NEW, die              ebenfalls die Dateien schliessen).

             PRINT #  verdichtet nicht die Daten auf der  Dis-             kette. Die Daten werden  so auf die Diskette  ge-

             schrieben, als wuerden sie auf dem Bildschirm mit              einer  PRINT-Anweisung  abgebildet.   Aus  diesem              Grund  sind  die Daten auf der  Diskette  zu  be             grenzen,  damit  sie  korrekt  von  der  Diskette              wieder eingelesen werden koennen.

             Beispiel:

             In der Liste der Ausdruecke sind numerische  Aus             druecke durch Semikolon zu trennen:

                      PRINT #1, A;B;C;D

             Werden   Kommas  verwendet,  werden die  zusaetz             lichen Leerstellen, die zwischen den Druckfeldern              eingesetzt  werden,  auch  auf die  Diskette  ge-             schrieben.

             Zeichenkettenausdruecke  muessen durch  Semikolon              in der Liste getrennt werden.

             Um das Format der Zeichenkettenausdruecke richtig              auf  der  Diskette  aufzuteilen,   verwendet  man              spezielle  Trennzeichen  in der  Liste  der  Aus             druecke.

             Beispiel:          A$="KOMBINAT" : B$="ROBOTRON"

             Die Anweisung      PRINT #1,A$;B$

             wuerde  auf die Diskette    

                                KOMBINATROBOTRON              schreiben.

             Weil  so keine Trennzeichen aufgezeichnet werden,              muss dieses Beispiel wie folgt programmiert  wer             den:

                                PRINT #1,A$;",";B$ 

             Nun wird auf die Diskette aufgezeichnet:

                                KOMBINAT,ROBOTRON

             Dieser Satz kann  in  zwei  Zeichenkettenvariable              wieder eingelesen werden.

             Wenn die Zeichenketten selbst  Komma,  Semikolon,              bedeutsame vorausgehende Leerstellen, Wagenrueck             lauf  oder  Zeilenschaltung  enthalten,  schreibt               man  sie auf die Diskette von  Anfuehrungszeichen              umgeben = CHR$(34).

             Beispiel:            A$="MEIER,WERNER":

                                  B$="   KARL-MARX-STADT"

             Die Anweisung      PRINT #1,A$;B$

             wuerde  den  folgenden  Satz  auf  die   Diskette              schreiben:

                   MEIER,WERNER   KARL-MARX-STADT

             und die Anweisung    INPUT #1,A$,B$     wuerde 

                 "MEIER" zu A$    und 

                 "WERNER   KARL-MARX-STADT"    zu B$ eingeben.

             Um  diese Zeichenketten richtig auf der  Diskette              zu trennen,  werden Anfuehrungszeichen unter Ver             wendung von CHR$(34) auf die Diskette ausgegeben.

             Die Anweisung 

                  PRINT #1,CHR$(34);A$;CHR$(34);CHR$(34);

                  B$;CHR$(34)

             schreibt den folgenden Satz auf die Diskette:

                  "MEIER,WERNER""   KARL-MARX-STADT"

             und die Anweisung      INPUT #1,A$,B$

             wuerde "MEIER,WERNER"       zu A$ 

             und    "   KARL-MARX-STADT" zu B$  eingeben.

             Die  PRINT #-Anweisung  kann auch mit  der  Wahl-             moeglichkeit  USING  verwendet  werden,   um  das              Format auf der Diskettendatei zu kontrollieren.

               Beispiel: PRINT #1, USING "####.##,";A,B,C,D

             Das  Komma am Ende der  Formatzeichenkette  dient              dazu,  die Daten auf der Diskette zu trennen.  Es              ist   somit  nicht   erforderlich,   zusaetzliche              Trennzeichen aufzuzeichnen.

5.3.2._WRITE_#
Format:      WRITE #<Dateinummer>, <Liste der Ausdruecke>

ZWECK:       Ausgabe von Daten in eine sequentielle Datei.

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              im "O"-Modus eroeffnet wurde.

             Die  <Ausdruecke> in der Liste sind Zeichenfolgen              oder  numerische  Ausdruecke.  Sie muessen  durch              Komma getrennt werden.

             Der Unterschied zwischen WRITE # und PRINT #  ist              der,  dass WRITE # Kommas zwischen die Datenworte              setzt,  wenn  sie  auf die  Diskette  geschrieben              werden,   und  Zeichenfolgen  durch  Anfuehrungs             striche abgrenzt. Damit wird es fuer den Anwender              nicht  erforderlich,  Trennzeichen  in die  Liste              einzufuegen.

             Eine  Wagenruecklauf- /Zeilenschaltungsfolge wird              eingefuegt,  nachdem das letzte Datenwort in  der              Liste auf die Diskette geschrieben wurde.

Beispiel:           A$="KOMBINAT": B$="ROBOTRON"

             Die Anweisung       WRITE #1,A$,B$

             schreibt das folgende Bild auf die Diskette:

                    "KOMBINAT","ROBOTRON"

             Eine nachfolgende INPUT #-Anweisung wie z.B.

                    INPUT #1,A$,B$     wuerde

                    "KOMBINAT" auf A$  und

                    "ROBOTRON" auf B$  eingeben.

5.3.3._INPUT_#
Format:      INPUT # <Dateinummer>, <Variablenliste>

Zweck:       Einlesen  von Daten aus einer sequentiellen  Dis             kettendatei und Wertzuweisung an Variable.

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              fuer die Eingabe eroeffnet wurde.

             <Variablenliste> enthaelt die Variablennamen, die              den  Positionen  in der Datei  zugeordnet  werden              sollen.

             Der  Variablentyp  muss zu dem  Typ  passen,  der              durch den Variablennamen spezifiziert wurde.

             Bei  INPUT # wird kein Fragezeichen angezeigt wie              bei INPUT.

             Die  Datenpositionen erscheinen in der Datei  so,              als wenn sie mit einer INPUT-Anweisung eingegeben              wuerden.

             Bei numerischen Werten werden vorangehende  Leer-             stellen,  Wagenruecklaeufe  und Zeilenschaltungen              ignoriert.

             Das erste Zeichen,  das keine Leerstelle,  keinen              Wagenruecklauf   und  keine  Zeilenschaltung  be             deutet, ist der Beginn einer Zahl.

             Die  Zahl  endet  an  einer   Leerstelle,   einem              Wagenruecklauf,  einer  Zeilenschaltung  oder  am              Komma.

             Wenn das BASIC-System die sequentielle Disketten             datei  nach einer Zeichenkettenposition  abfragt,              werden vorangehende Leerstellen, Wagenruecklaeufe              oder Zeilenschaltungen ebenfalls ignoriert.

             Das erste Zeichen, das keine  Leerstelle,  keinen              Wagenruecklauf,  keine  Zeilenschaltung bedeutet,              ist der Beginn einer Zeichenkette.

             Wenn dieses erste Zeichen ein  Anfuehrungszeichen              (")  ist,  besteht  die  Zeichenkette  aus  allen              Zeichen,  die  zwischen  dem ersten  Anfuehrungs             zeichen und dem zweiten gelesen  werden. Folglich              darf  eine Zeichenkette  kein  Anfuehrungszeichen              als Zeichen enthalten. 

             Ist  dieses erste Zeichen  der  Zeichenkette kein              Anfuehrungszeichen,  endet  die  Zeichenkette  an              einem  Komma,  Wagenruecklauf oder einer  Zeilen             schaltung (oder nach 255 Zeichen).

5.3.4._LINE_INPUT_# 

Format:      LINE INPUT # <Dateinummer>, <Zeichenketten-

                        variable>

Zweck:       Einlesen  einer ganzen Zeile (max.  254  Zeichen)              ohne  Trennzeichen  aus einer sequentiellen  Dis             kettendatei in eine Zeichenkettenvariable.

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              eroeffnet wurde.

             <Zeichenkettenvariable>  ist  der  Variablenname,              dem die eingelesenen Zeichen zugeordnet werden.

             LINE INPUT # liest alle Zeichen der sequentiellen              Datei  bis  zu einem  Wagenruecklauf.  

             Die  Eingabe wird nicht durch  Anfuehrungszeichen              oder Komma beendet, sondern nur durch Wagenrueck-             lauf.

5.4._Direktzugriffsdateien
Die  bei  Direktzugriffsdateien  verwendeten  Anweisungen  und Funktionen sind:

          FIELD

          LSET / RSET

          PUT

          GET

          MKI$, MKS$, MKD$

          CVI, CVS, CVD

          EOF

          LOC

          LOF

Die  folgenden  Programmschritte sind  erforderlich,  um  eine Direktzugriffsdatei_zu_schaffen:
                                        Beispiel:
1.  Eroeffnen (OPEN) der Datei im       OPEN "R",#1,"DATEI",32

    "R"-Modus.

    In diesem Beispiel wird eine

    Satzlaenge von 32 Byte vorge-

    geben. 

    Wenn die Satzlaenge nicht ange-

    geben wird, werden 128 Byte

    vorgesehen.

2.  Die FIELD-Anweisung ist zu ver-     FIELD #1,20 AS N$,

    wenden, um Platz fuer die Vari-     4 AS A$, 8 AS B$

    ablen im Direktzugriffspuffer

    zuzuordnen.

3.  Es ist LSET zu verwenden, um        LSET N$=X$

    die Daten in den Direktzugriffs-    LSET A$=MKS$(AMT)

    pufferspeicher zu bringen.          LSET B$=T$

    Numerische Werte muessen in

    Zeichenkettenform gebracht 

    werden, wenn sie in den Puffer

    uebernommen werden sollen. 

    Dafuer sind die "make"-Funk-

    tionen zu verwenden:

      MKI$, MKS$, MKD$

4. Die Daten werden mit der PUT-        PUT #1,CODE%

   Anweisung aus dem Puffer auf die

   Diskette geschrieben.

Anmerkung:  In  einer  INPUT- oder  LET-Anweisung  darf  keine             FIELD-Zeichenkettenvariable  verwendet  werden, da

            sonst der Zeiger (Pointer) fuer diese Variable  in             den   Zeichenkettenraum  zeigt,   anstatt  in  den             Pufferspeicher der Direktzugriffsdatei.

Die folgenden Programmschritte sind erforderlich,  um auf eine Direktzugriffsdatei_wieder_zuzugreifen:
1.  Eroeffnen (OPEN) der Datei "DATEI" im "R"-Modus

2.  Es ist die FIELD-Anweisung zu nutzen, um im Pufferspeicher

    fuer  den Direktzugriff Platz fuer die Variablen zu reser    vieren, die aus der Datei gelesen werden sollen:

               FIELD #1,20 AS N$, 4 AS A$, 8 AS B$

Anmerkung: In einem Programm, das sowohl Ein- als auch Ausgabe            mit  derselben  Datei vornimmt,  wird oft nur  eine            OPEN- und eine FIELD-Anweisung verwendet.

3.  Es ist die GET-Anweisung zu verwenden, um den gewuenschten     Satz in den Pufferspeicher fuer den Direktzugriff   einzu    lesen:     GET #1,CODE%

4.  Vom  Programm  kann nun auf die Daten  im  Pufferspeicher     zugegriffen werden. 

    Numerische  Werte  muessen  wieder unter  Verwendung  der     "convert"-Funktionen CVI,  CVS,  CVD in Zahlen verwandelt     werden.

5.  Die LOC-Funktion  bringt  bei  Direktzugriffsdateien  die 

    Nummer des laufenden Satzes zurueck.

    Die Nummer des laufenden Satzes ist 1 plus die Nummer des

    letzten Satzes, der in einer GET- oder PUT-Anweisung ver-

    wendet wurde.

6.  Die EOF-Funktion wird wie bei sequentiellen Dateien  ver-    wendet.

7.  Die LOF-Funktion bringt die Anzahl der Saetze zurueck, die         im letzten geschriebenen oder gelesenen  Bereich  (Extent)         = 16 K vorhanden sind.    Sie  ist  vor  der  Anweisung  CLOSE  aufzurufen,  um  den     aktuellen  Satzzaehler  der Datei (Nummer des letzten  ge-    schriebenen Satzes) zu ermitteln.

    Nachfolgend  muss auf den letzten Satz der Datei  mit  GET     positioniert werden, um das tatsaechliche Ende der bereits     existierenden Datei zu erreichen.

    Erst dann kann  die Datei geschlossen werden (CLOSE).

Alle  Details sind den jeweiligen Anweisungen oder  Funktionen bzw. Beispielen zu entnehmen.

5.4.1._FIELD
Format:      FIELD [#]<Dateinummer>, <Feldbreite> AS

                      <Zeichenkettenvariable> ...

Zweck:       Reserviert  Speicherplatz fuer einen  Puffer,  in              dem   Variable   fuer  eine   Direktzugriffsdatei              abgelegt werden.

Bemerkungen: Bevor  eine GET- oder  PUT-Anweisung  ausgefuehrt              werden    kann,    muss   eine    FIELD-Anweisung              programmiert  werden,  um den Pufferspeicher fuer              den Direktzugriff einzustellen.

             Die maximale Pufferlaenge betraegt 128 Zeichen.

             <Dateinummer> ist die Nummer, unter der die Datei              eroeffnet wurde.

             <Feldbreite>  ist  die Anzahl  der  Zeichen,  die              einer <Zeichenkettenvariablen> zugeordnet  werden              soll.

             Zum Beispiel ordnet die Anweisung

                 FIELD 1, 30 AS B$, 10 AS ID$, 50 AS AD$

             die ersten 30 Positionen (Byte) im Pufferspeicher              fuer den Direktzugriff der Zeichenkettenvariablen              B$  zu,  die naechsten 10 Positionen ID$ und  die              naechsten 50 Positionen AD$.

             Die  Daten werden erst mit den  Anweisungen  LSET              und RSET in den Puffer eingeschrieben.

             Im  Puffer  koennen  nur  Zeichenkettenwerte  ge-             speichert  werden.   Deshalb  muessen  numerische              Werte,  bevor sie in den Puffer geschrieben  wer             den,  erst in Zeichenkettenwerte konvertiert wer             den.

             Die  Gesamtzahl  der Bytes,  die in einer  FIELD-             Anweisung zugeordnet werden,  darf die  Aufzeich             nungslaenge  nicht uebersteigen,  die  festgelegt              wurde,  als  die  Datei eroeffnet  wurde.             Sonst tritt ein Fehler "Field overflow" auf.

             Fehlt   die   Angabe   der   Aufzeichnungslaenge,              betraegt diese 128 Byte.

             Es   kann   eine  beliebige  Anzahl  von   FIELD-             Anweisungen  fuer die gleiche  Datei  ausgefuehrt              werden.  

 Anmerkung:  Verwenden_Sie_keinen_FIELD-Variablennamen_in 

             einer_INPUT-_oder_LET-Anweisung.

             Wenn  einmal ein Variablenname  ein  Speicherfeld              belegt,  weist er auf den korrekten Platz in  dem              Pufferspeicher  fuer den Direktzugriff hin.  Wird             eine  nachfolgende INPUT- oder LET-Anweisung  mit              jenem Variablennamen ausgefuehrt,  wird der Vari             ablenzeiger zum Zeichenkettenplatz bewegt.

 Beispiele:    10 OPEN "R",#1,"A:TELEFON",35

               20 FIELD #1,2 AS NR$, 33 AS LEER$

               30 FIELD #1,25 AS NAMEN$, 10 AS TELE$

             Dieses Beispiel zeigt das mehrfache Verwenden der              FIELD-Anweisung.

               10 FOR LOOP% = 0 TO 7

               20 FIELD #1,(LOOP% * 16) AS OFFSET$, 

                  16 AS A$(LOOP%)

               30 NEXT LOOP%

             Hier  wird das Verwenden der FIELD-Anweisung  ge-             zeigt  fuer  eine Anzahl von  Elementen  gleicher              Groesse.

             Das Resultat ist gleich den einzelnen Befehlen:

               FIELD #1,16 AS A$(0). 16 AS A$(1), ...

                        16 AS A$(6), 16 AS A$(7)

5.4.2._LSET__und__RSET
Format:      LSET <Zeichenkettenvariable> = <Zeichen-

                  kettenausdruck>

Zweck:       Bewegt  Daten aus dem Speicher zu  einem  Puffer             speicher  fuer den Direktzugriff (in Vorbereitung              einer PUT-Anweisung).

Bemerkungen: Wenn  <Zeichenkettenausdruck> weniger Byte erfor             dert,  als  der <Zeichenkettenvariablen>  in  der              FIELD-Anweisung  zugeteilt wurde,  justiert  LSET              die  Zeichenkette  links im Bereich und RSET  die              Zeichenkette rechts im Bereich.

             Leerraeume  werden  verwendet,   um  zusaetzliche              Positionen anzupassen.

             Ist die Zeichenkette zu lang fuer das Feld,  wer             den Zeichen rechts abgeschnitten.

             Numerische    Werte    sind   in    Zeichenketten              umzuwandeln,  ehe die Anweisungen LSET oder  RSET              wirken (siehe die Funktionen MKI$, MKS$, MKD$).

Beispiel:      LSET A$=MKS$(ZAHL)

               LSET B$=ZEICHEN$

Anmerkung:   LSET  oder  RSET koennen auch bei einer  Zeichen-             kettenvariablen  ohne  FIELD-Anweisung  verwendet              werden (keine Dateiarbeit),  um eine Zeichenkette              in  einem gegebenen Bereich links oder rechts  zu              justieren.

             Die Programmzeilen z. B. 

                         50 L$=LEER$(20)

                         60 RSET L$=Z$

             justieren  die Zeichenkette rechts in einem  Feld              von 20 Zeichen.

5.4.3._PUT
Format:      PUT [#]<Dateinummer> [,<Satznummer>]

Zweck:       Schreibt  einen  Satz aus einem Puffer  fuer  den              Direktzugriff  in eine  Direktzugriffs-Disketten             datei.

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              eroeffnet wurde.

             Wenn  die <Satznummer> fehlt,  wird die  naechste              zur   Verfuegung  stehende  Satznummer  verwendet              (nach dem letzten PUT).

             Die hoechste Satznummer ist 32767, die niedrigste              Satznummer ist 1.

             PUT # schreibt die Daten in einen  Ausgabepuffer.              Ausgegeben werden die Daten erst, wenn der Puffer              voll  ist,  oder bei einem CLOSE  (bzw.  END oder              NEW, die ebenfalls die Dateien schliessen).

             Jeder  Versuch,  ueber die Grenze des Puffers  zu              lesen oder zu schreiben,  verursacht einen Fehler                   "Field overflow".

5.4.4._GET
Format:      GET [#]<Dateinummer> [,<Satznummer>]

Zweck:       Liest einen Satz von einer Direktzugriffsdatei in              einen Puffer fuer den Direktzugriff.

Bemerkungen: <Dateinummer> ist die Nummer, unter der die Datei              eroeffnet wurde.

             Wird  die <Satznummer> nicht angegeben,  wird der              naechste Satz (nach dem letzten GET) in den  Puf             fer eingelesen.

             Die groesstmoegliche Satznummer ist 32767.

5.4.5._Konvertierung_numerischer_Werte_in_Zeichenkettenwerte
Format:      MKI$(<Integerausdruck>)

             MKS$(<Ausdruck mit einfacher Genauigkeit>)

             MKD$(<Ausdruck mit doppelter Genauigkeit>)

Zweck:       Konvertiert  numerische Werte  in  Zeichenketten-             werte.

             Jeder numerische Wert, der im Dateipuffer mittels              der  LSET- oder RSET-Anweisung eingetragen  wird,              muss in eine Zeichenkette konvertiert werden.

             MKI$ setzt einen Integerwert in eine Zeichenkette              der Laenge 2 um.

             MKS$  bzw.  MKD$  konvertieren  einen  Zahlenwert              einfacher  bzw.  doppelter  Genauigkeit  in  eine              Zeichenkette von 4 bzw. 8 Byte Laenge.

Beispiel:      50 FIELD #1,8 AS D$, 4 AS N$

               60 LSET D$=MKD$(A#)

               70 LSET N$=MKS$(N)

               80 PUT #1

5.4.6. Konvertierung von Zeichenkettenwerten in numerische _______Werte______________________________________________
Format:      CVI(X$)

             CVS(X$)

             CVD(X$)

Zweck:       Konvertiert   Zeichenkettenwerte  in   numerische              Werte.

             Numerische Werte,  die von der Diskettendatei bei              Direktzugriff eingelesen werden,  muessen von der              Zeichenkettenform  in die numerische Form  gewan             delt werden.

             CVI konvertiert eine Zeichenkette der Laenge 2 in              einen Integerwert. 

             CVS bzw. CVD wandeln eine Zeichenkette der Laenge              4 bzw. 8 in einen numerischen Wert einfacher bzw.              doppelter Genauigkeit.

Beispiel:       50 FIELD #1,4 AS N$, 8 AS A$, 2 AS I$

                60 GET #1

                70 N=CVS(N$)

                80 A#=CVD(A$)

                90 I%=CVI(I$)

5.5._Funktionen
5.5.1._EOF
Format:      EOF(Dateinummer)

Zweck:       Gibt wahr (-1) aus,  wenn das Ende einer  sequen             tiellen oder Direktzugriffsdatei erreicht wurde.

             Wird nach Erkennen EOF erneut auf die Datei zuge             griffen,  wird  EOF solange erkannt,  bis ein er             neutes OPEN (nach CLOSE) auf die Datei erfolgt.

             Mit  der Funktion EOF koennen Fehler "Input  past              end" vermieden werden.

             <Dateinummer> ist die gleiche,  wie bei OPEN ver-             einbart.

Beispiel:      10 OPEN "I",1,"DATEN"

               20 C=0

               30 IF EOF(1) THEN 100

               40 INPUT #1, A(C)

               50 C=C+1 : GOTO 30

5.5.2._LOC
Format:      LOC(<Dateinummer>)

Zweck:       Bei Direktzugriffsdateien bringt LOC die naechste              Satznummer zurueck,  die zu benutzen ist,  um ein              GET oder PUT auszufuehren.

             Wurde die Datei eroeffnet, aber es hat noch keine              Disketten-Ein- /Ausgabe stattgefunden, bringt LOC              = 0/ zurueck.

             Bei sequentiellen Dateien gibt LOC die Anzahl der              Sektoren  aus  (128-Byte-Bloecke),  die  von  der              Datei  gelesen  oder  in  die  Datei  geschrieben              wurden, seit die Datei eroeffnet wurde.

Beispiel:       200 IF LOC(1) > 100 THEN STOP

5.5.3._LOF                

Format:      LOF(<Dateinummer>)

Zweck:       Bringt  bei Direktzugriffsdateien die Anzahl  der              Saetze  zurueck,  die im letzten  gelesenen  oder              geschriebenen Bereich (Extent) vorhanden sind.

             Ueberschreitet   die  Datei  einen  Bereich  (128              Saetze  = 16 K) nicht,  dann bringt LOF die  tat             saechliche Laenge der Datei zurueck.

             Die LOF-Funktion wird verwendet, um den aktuellen              Satzzaehler  bei  Zugriff auf  die  Datei  einzu-             stellen (s. Pkt. 5.4., 7.).

Beispiel:      110 IF NUM% > LOF(1) THEN PRINT "FEHLER"

5.5.4._INPUT$
Format:      INPUT$(X,[#]Y)

Zweck:       Bringt  eine Zeichenkette von X Zeichen  zurueck,              die von der Datei Y gelesen wurden.

Beispiel:      10 OPEN "I",1,"DATEN"

               20 IF EOF(1) THEN 50

               30 PRINT HEX$(ASC(INPUT$(1,#1)));

               40 GOTO 20

               50 PRINT

               60 END

5.5.5._VARPTR
Format:      VARPTR(#<Dateinummer>)

Zweck:       Bei  sequentiellen Dateien wird die  Startadresse              des Disketten- Ein- / Ausgabe-Puffers zugewiesen.

             Bei  Direktzugriffsdateien  wird die Adresse  des              FIELD-Puffers der <Dateinummer> zugewiesen.

5.6._Beispiele
5.6.1._Sequentieller_Zugriff
         5 REM BEISPIEL 1

         10 OPEN "O",#1,"PROGR1.ASC"

         20 INPUT "NAME:  ";N$

         30 IF N$="E" THEN GOTO 80

         40 INPUT "ADRESSE:  ";A$

         50 PRINT #1,N$;",";A$

         60 PRINT

         70 GOTO 20

         80 CLOSE #1

         90 OPEN "I",#1,"PROGR1.ASC"

         100 INPUT #1,NI$,AI$

         110 PRINT NI$,AI$

         120 IF EOF(1) THEN END

         130 GOTO 100

         5 'BEISPIEL 2

         10 'ERWEITERN DATEI "ARTIKEL.ASC"

         20 PRINT CHR$(12);: IC=1  

         30 PRINT "PROGRAMM ERWEITERT EINE ARTIKELDATEI" 

         40 PRINT "===================================="

         50 OPEN "I",#1,"ARTIKEL.ASC"

         60 OPEN "O",#2,"TEMPOR.ASC"

         70 LINE INPUT #1, N$

         80 IF N$="$$$" THEN 90 ELSE PRINT #2,N$: IC=IC+1

         85 INPUT #1,P: PRINT #2,P: GOTO 70

         90 CLOSE #1

         100 PRINT

         110 PRINT TAB(38);"|---------------------------|"

         120 PRINT "(";IC;")  ";

         125 LINE INPUT "ARTIKELNAME (30 Z max.) : ",N$

         130 IF N$="" THEN 170

         140 IF LEN(N$)>30 THEN 120

         150 INPUT "PREIS : ";P

         160 IC=IC+1:PRINT:PRINT #2,N$:PRINT #2,P: GOTO 110

         170 PRINT #2,"$$$":CLOSE #2

         180 KILL "ARTIKEL.ASC"

         190 NAME "TEMPOR.ASC" AS "ARTIKEL.ASC"

         200 END

5.6.2._Direktzugriff
         5 REM BEISPIEL 1

         10 OPEN "R",#1,"PROGR2.ASC",70

         15 SATZ=2

         20 FIELD #1,20 AS N$,50 AS A$

         30 INPUT "NAME:   ";N1$

         40 IF N1$="E" THEN GOTO 100

         50 INPUT "ADRESSE:";A1$

         60 LSET N$=N1$

         70 LSET A$=A1$

         80 PUT #1,SATZ

         85 SATZ=SATZ+2

         90 GOTO 20

         100 CLOSE #1

         110 END

         5 REM BEISPIEL 2

         10 OPEN "R",#1,"PROGR2.ASC",70

         20 SATZ=2

         30 FIELD #1,20 AS N$, 50 AS A$

         40 GET #1, SATZ

         60 PRINT N$,A$

         70 IF EOF(1) THEN END

         80 SATZ=SATZ + 2

         90 GOTO 30

         100 END

         5 REM BEISPIEL 3 (LAUFWERK B:)

         10 OPEN "R",#1,"B:PROGR3.ASC"

         15 SATZ=2

         20 FIELD #1,20 AS N$, 50 AS A$, 5 AS T$

         30 INPUT "NAME:    ";N1$

         40 IF N1$="E" THEN GOTO 100

         50 INPUT "ADRESSE: ";A1$

         55 INPUT "TELEFON: ";T%

         60 LSET N$=N1$

         70 LSET A$=A1$

         75 LSET T$=MKI$(T%)

         80 PUT #1, SATZ

         85 SATZ= SATZ+2

         90 GOTO 20

         100 SATZ=2

         105 GET #1, SATZ

         110 SATZ=SATZ+2

         120 PRINT N$, A$, CVI(T$)

         130 IF EOF(1) THEN END

         140 GOTO 105

                          - # -


